How to build a cube and animate it with MAT’s Scasm Editor

Let me try to start with an explanation how I “see” the 3D world. In my way to design things I follow an own way which comes from the hours of working with objects.

So don’t be confused if others do some things in a different way.

Let’s start with this:

I imagine a cube which is in front of me.

The cube has a center. This is what we call the Reference Point. 

To make it a bit easier let’s say the Reference Point is the middle of the bottom side of the cube. The cube shall be 40 units width. 

And this is the result: A really complicate cube, isn’t it? ;-)

The read lines are the axis of the 3D world. In other words: East is represented by the positive values of the X axis, West by the negative a.s.o. In this case we don’t see the negative values of the Z axis. The RefPoint is where the 3 axis meet in the middle of the bottom. 

If ( IF ! ) we would decide to have the RefPoint in the middle of the cube, we would have to raise the point where the axes are meeting upwards.

Now let’s see, what the red numbers means. The numbers define the outer points of the cube.

From bottom right to top right of the front square I count 0,1,2,3

Others count different but I use this system.

Now let’s see, which values represent this points.

Point 0 is represented by  
x  20  (you don’t need the + sign in this case )





z    0 





y –20

Try to find out the location of the other three points!

Now we need 4 more point to have the points of the cube. It’s 4 and 5 on the left side and 6, 7 on the right side.

Now let’s check, which points represent the top of the cube.

From my point of view and from right to left they are:

3, 2 , 5 , 7

To see which points represent the back of the cube I imagine to walk around it…

Now it is 4, 6 , 7 , 5

Have in mind, that I count from right to left.

To know this is important if you have to lay the textures over the cube. You need to define from which point to which point the texture has to be painted.

Now let’s do the cube:

Start the Scasm Editor and select the “Normal object template”. Select “Write to source” and close this window.

Search for the following entries in the main Editor window and place the cursor below the ;  --- > !!!
:Start

; ------>  !!!      Place your objectcode here !!! points and polygons

Return

Now select the PointList command.

Do you remember how many points we need for the cube??

Set the number of points and use “Simulate”. In your main window you should see:

Points( 0    ; Start of pointlist 

    0   0   0  ;  point 0

    0   0   0  ;  point 1

    0   0   0  ;  point 2

    0   0   0  ;  point 3

    0   0   0  ;  point 4

    0   0   0  ;  point 5

    0   0   0  ;  point 6

    0   0   0  ;  point 7

 ) ; End of pointlist

Note: We start with 0 so we only need 7 more points to have all 8.

This is only a template. Now we have to fill the coordinates manually. Others do that by using a tool like VOD or FSDS, I have only used that in the beginning. There are no tools which are able to let me create my objects in the way I want them.

And believe me: If you want to work with the advanced possibilities of Scasm you need to know exactly where every point of an object is located. How else do you want to decide which parts of an object rotates around which axis? Later I will show you why I thinbk this is necessary in an example.

It just a question of practicing and you are able to name every point of your cat and know the coordinates ;-)

Ok, what we need is this:

Points( 0

      20  0   -20   ; 0 right bottom  front

     -20  0   -20   ; 1 left  bottom  front

     -20 40   -20   ; 2 left  top     front

      20 40   -20   ; 3 right top     front

     -20  0    20   ; 4 left  bottom  back

     -20 40    20   ; 5 left  top     back

      20  0    20   ; 6 right bottom  back

      20 40    20   ; 7 right top     back

      )

Ok, if you are not familiar with the 3D world try to locate every point and check it’s coordinates, and write them to your pointlist template. Otherwise you may copy the pointlist above and paste it to the empty one. 

Now we should add some textures to our cube.

Let’s select the bitmap first.

Place the cursor in the main window below the bracket of our pointlist and select the “LoadBimap” command from the Editors menue.

Click on “Load Bitmap from drive” and select the cube1.bmp.

Now you should see this incredible red masterpiece:

You see, that the Editor already has written the basic command for you. Now select if you want a night bitmap ( you will need an existing cube1_LM.bmp !!) and set the fallback colors for the case the texture is not available.

Then click on “Write to source”.

Now we have to tell how the loaded texture has to be painted to the cube.

We do this by using the TexPoly command from the Editor:

This is a real basic tool and I will have to work on it. Right now let’s just use the “Fasthack” button and write the code to the Editors main window.

It should appear below the LoadBitmap command!

What you should get is something like this:

LoadBitmap( 0 L6 EF 0xFF 0xFF 0x00 cube1.bmp ) ; used bitmap: cube1.bmp additional bitmaps: 

TexPoly( a

0   0   0  ; point 0  front

1   255   0  ; point 1

2   255   255  ; point 2

3   0   255  ; point 3

 )  ; End of TexPoly

What you see is, that this command will start at point 0 0 of the bitmap and starts drawing at the point 0 of our cube. Then it draws to point 255 0 of our bitmap and draws this to point 1 of our cube.

Easy, isn’t it?

But now let’s see how we cover the left side of our cube.

Use the same TexPoly command ( I use to copy and paste it ).

The points of the bitmap are still the same but we need to draw them to different points of the cube.

The left side is represented by point 1, 4, 5, 2

Again: Have in mind that I use to start at the bottom right and count clockwise…

Simply substitute the point numbers as below:

TexPoly( a

1   0   0  ; point 1          left side right bottom

4   255   0  ; point 4      left side left bottom

5   255   255  ; point 5  left side left top

2   0   255  ; point 2      left side right top

 )  ; End of TexPoly
The same procedure for the top:

TexPoly( a

3   0   0  ; point 0

2   255   0  ; point 1

5   255   255  ; point 2

7   0   255  ; point 3

 )  ; End of TexPoly

Ok, now you should test your object in a scenery. Save your template and place it with Airport. ( You should know how to do that).

Please have in mind that there are a lot of more types of commands in Scasm. As I wrote: you should be familiar with it, this is an example how to create rotation, not “how to create 3D objects”.

If you want you may use the cubeexa1.api which comes with the editor. It creates the cube.

Now let’s rotate the thing.

We want to start with a “normal rotation”. Open a new “normal header template” in MAT’s Scasm Editor and write it to the main window.

Set the cursor below the “:Start” label and call the Rotation Type I command.

Let’s use “cuberotation” as label and “cube” as object.

Now let’s have a look at the values. The delta values would shift our cube from it’s RefPoint. Remember: the Refpoint is in the middle of the bottom at the time. We will see a bit later how to work with the setting of the Refpoint. At the time we will leave the values at 0.

The “deg” values represent the rotation of the object at the time when the rotation starts. At the time we have the default heading, pointing north. Let’s turn the cube 45 deg to east. The axis we need to rotate around  is the Z axis. Lets set the value to 45 (positive) and simulate the coding.

You will get:

:cuberotation

TransformCall( :cube 0 0 0 0 0 45 0 0 0 )    ; Rotated object = :cube

Return

The TransformCall is calling the object  at the :cube label and turns it 45 deg.

We need to add our cube to the coding.

Save your code as  cuberot1.api or better  open another instance of the Scasm Editor and load your cube api there. Now copy the code between the :Start label

and the Return which draws our cube.

Insert a new label in our source code, the label needs a Return command to be terminated:

:cuberotation

TransformCall( :cube 0 0 0 0 0 0 0 45 0 )    ; Rotated object = :cube

Return
:cube

Return

Now paste the code of the cube between the label and the return.

From the :Start label down to the end of the code it should now look like this:

:Start

:cuberotation

TransformCall( :cube 0 0 0 0 0 0 0 45 0 )    ; Rotated object = :cube

Return

:cube

    ;  X  Z    Y

Points( 0

      20  0   -20   ; 0 right bottom  front

     -20  0   -20   ; 1 left  bottom  front

     -20 40   -20   ; 2 left  top     front

      20 40   -20   ; 3 right top     front

     -20  0    20   ; 4 left  bottom  back

     -20 40    20   ; 5 left  top     back

      20  0    20   ; 6 right bottom  back

      20 40    20   ; 7 right top     back

      )

LoadBitmap( 0 L6 0xEF 0xFF 0xFF 0x00 cube1.bmp ) ; used bitmap: cube1.bmp additional bitmaps: cube1_LM.bmp

; Frontside of the cube

TexPoly( a

0   0   0  ; point 0  front

1   255   0  ; point 1

2   255   255  ; point 2

3   0   255  ; point 3

 )  ; End of TexPoly

LoadBitmap( 0 6 0xEF 0xFF 0xFF 0x00 cube2.bmp ) ; used bitmap: cube1.bmp additional bitmaps: cube1_LM.bmp

;Left side of the cube

TexPoly( a

1   0   0  ; point 0 left side

4   255   0  ; point 1

5   255   255  ; point 2

2   0   255  ; point 3

 )  ; End of TexPoly

;Top of the cube

TexPoly( a

3   0   0  ; point 0

2   255   0  ; point 1

5   255   255  ; point 2

7   0   255  ; point 3

 )  ; End of TexPoly

LoadBitmap( 0 L6 0xEF 0xFF 0xFF 0x00 cube1.bmp ) ; used bitmap: cube1.bmp additional bitmaps: cube1_LM.bmp

;Backside of the cube

TexPoly( a

4   0   0  ; point 0

6  255   0  ; point 1

7   255   255  ; point 2

5   0   255  ; point 3

 )  ; End of TexPoly

LoadBitmap( 0 6 0xEF 0xFF 0xFF 0x00 cube2.bmp ) ; used bitmap: cube1.bmp additional bitmaps: cube1_LM.bmp

;Right side of the cube

TexPoly( a

6   0   0  ; point 0

0  255   0  ; point 1

3   255   255  ; point 2

7   0   255  ; point 3

 )  ; End of TexPoly

;Bottom of the cube

TexPoly( a

6   0   0  ; point 0

4  255   0  ; point 1

1   255   255  ; point 2

0   0   255  ; point 3

 )  ; End of TexPoly

Return

Return

EndA

Now save your api and use Airport to place it somewhere and check it. The cube is turned 45 deg. This is his starting position.

Now go back to The Scasm Editor and select the variable 30A for the Z axis. This variable will turn our object. Substitute the 

:cuberotation

TransformCall( :cube 0 0 0 0 0 0 0 45 0 )    ; Rotated object = :cube

Return
with:

:cuberotation

TransformCall( :cube 0 0 0 0 0 0 0 45 30A )    ; Rotated object = :cube

Return
Hmmm… you could do that also by changing the code manually and add the 30A.

Now compile your scenery again and watch the cube rotating.

Easy stuff, isn’t it? But let’s check what we can do with this command:

· we could change the RefPoint

· we could set a start point for the rotation

· we can select a variable to turn the object.

But the thing is always turning clockwise. And it’s turning always in the same direction. Always with the same speed. How boring…..

That’s the reason why we should have a look at the “advanced” rotation.

Before we start you should also check what happens if you move the RefPoint of the X  axis in the TransformCall. 

Advanced Rotation / Movement

Example: cuberot1b.api
Now let’s see what “advanced” means.

First create your template with the cube again or delete the cuberotation part to make space for the new one.

Call the 2a Rotation / Movement commandv from the menue. Set the values as shown below:

We have called the new label to create “cuberotation” and call our object by the labelname “cube”) .

We have set the loop to 60 steps which means 1 step per sec. – the 30A is running from 0 up to 65534 in 60 seconds.

We have selected the TransformCall as command. 

Note: You don’t really need the implemented RotatedCall, everything could be done by using the TransformCall – even a simple rotation.

Now we have set the start value for delta Z to 200. This means the RefPoint of the cube is shifted up to 200. 

Then we have set the moving for delta z to 3 per loopstep. This means the delta Z will be increased for 3 points every loopstep. 

The loop was set to “reversed”. 

This means: The Z will start at 200, goes 30 steps with an increment of 3 upwards and 30 steps with an decrement of 3 downwards.

Simple said: this values move the cube upwards and downwards like a lift.

The command looks like this:

:Step0

IfVarRange( :Step1 30A 0 1074 )

TransformCall( :cube 0 200+0 0 00 0 00 0 00 0 )

Return

We start with step 0 . In this step we see the delta z at 200.

:Step1

IfVarRange( :Step2 30A 1075 2149 )

TransformCall( :cube 0 200+3 0 0+2 0 0+2 0 0+2 0 )

Return

In step 1 we see the delta z at 200+3 and so on...
:Step2

IfVarRange( :Step3 30A 2150 3224 )

TransformCall( :cube 0 200+6 0 0+4 0 0+4 0 0+4 0 )

Return

Now let’s check what the other settings mean:

All three axes of the cube are rotated for 2 points per loopstep and the rotation goes backward on all three axes after the middle of the loop.

Check how it looks if you animate the cube in this way.

Note: This example is only for demonstration. You don’t need to use reversed loops but the whole technic has one drawback: If you don’t reverse the loop the object jumps back to it’s startvalues after reaching the last step ( after 1 minute ).

I would recommend to use this technic only for rotation and movements which go back to the starting point. My searchlight are done with such a type of loop. They move over two axes, they turn and the beam goes up and down.

We will implement a better way ( longer running loops ) for movement to the Editor later.

One more thing: The “DoubleLoop” command is very helpful if you want to rotate ( or move ) two objects in the same loop. Imagine an object with a base rotating left and right and something on top which is moving up and down and reversed in the same time. You could also do that with two single loops but it helps to keep the code small.

Happy experiments!

